

Program: SF-110\_SampleProgram.CR1

```
'CR1000
'Program for measuring Apogee Instrument's radiation frost detection sensor (SF-110)
'Author: Jacob Bingham
'Date: 18-Apr-2012
'
'Wiring Diagram:
' EX1 (Red) ----- Excitation for leaf thermistor = 2.5V
'
' SE1 (Black) ----- Single-ended channel (positive leaf thermistor lead)
'
' AG (Blue) ----- Analog ground (negative leaf thermistor lead)
'
' EX1 (White) ----- Excitation for bud thermistor = 2.5V
'
' SE2 (Green) ----- Single-ended channel (positive bud thermistor lead)
'
' AG (Orange) ----- Analog ground (negative bud thermistor lead)
'
'Explanation of variables used in program
'PanelT = datalogger panel temperature
'BattV = datalogger battery voltage
'LeafTempC = leaf temperature in degrees Celsius
'LeafTempK = leaf temperature in Kelvin
'LeafTempF = leaf temperature in degrees Fahrenheit
'BudTempC = bud temperature in degrees Celsius
'BudTempK = bud temperature in Kelvin
'BudTempF = bud temperature in degrees Fahrenheit

'Declare public variables
Public PanelT, BattV
Public LeafTempC, LeafTempK, LeafTempF
Public BudTempC, BudTempK, BudTempF

'Define data table (table is outputting data every 60 seconds)
DataTable (SF-110,1,-1)
DataInterval (0,60,Sec,10)
Minimum (1,BattV,FP2,0,False)
Sample (1,PanelT,FP2)
Average (1,LeafTempC,FP2,False)
Average (1,LeafTempK,FP2,False)
Average (1,LeafTempF,FP2,False)
Average (1,BudTempC,FP2,False)
Average (1,BudTempK,FP2,False)
Average (1,BudTempF,FP2,False)
EndTable

'Main program
BeginProg
Scan (5,Sec,0,0)

'Measure Battery Voltage and Panel Temperature
Battery (BattV)
PanelTemp (PanelT,_60Hz)

'Use Therm109 command to calculate leaf and bud temperatures
Therm109 (LeafTempC,1,1,Vx1,0,_60Hz,1.0,0)
Therm109 (BudTempC,1,2,Vx1,0,_60Hz,1.0,0)

'Instructions to calculate temperature in Kelvin and degrees Fahrenheit
LeafTempK = LeafTempC + 273.15
LeafTempF = LeafTempC * 1.8 + 32
BudTempK = BudTempC + 273.15
```

Program: SF-110\_SampleProgram.CR1

```
BudTempF = BudTempC * 1.8 + 32
```

```
'Call output table
```

```
CallTable SF-110
```

```
NextScan
```

```
EndProg
```